

# **Graph Macro Tools You Can Use in SAS® V9.3, With or Without SAS/GRAPH®**

LeRoy Bessler PhD, Bessler Consulting and Research, Le\_Roy\_Bessler@wi.rr.com

## **ABSTRACT**

My quest since 1981 for communication-effective graphs has yielded some solutions that require extra coding. Macros can do that coding work for you, and give you simple parameters to specify instead, similar to using a SAS procedure. I share and show how to use three such macros for common uses with high-value results.

Two of the macros overcome limitations in the new ODS GRAPHICS and SG procedures which are now an alternative to SAS/GRAPH. With those macros you need not use SAS/GRAPH to compensate for what is missing in the new technology. Their output charts are written to disk to later be inserted into Microsoft PowerPoint, Word, or Excel, or simply printed, but they could be web-enabled and packaged with ODS HTML.

The third macro does require use of SAS/GRAPH and creates web-enabled output. However, its image file can be manually inserted into Microsoft PowerPoint, Word, or Excel, or simply printed.

**The macros that rely on ODS GRAPHICS and SG procedures require Version 9.3 of Base SAS software. The macro that relies on SAS/GRAPH works fine in both Version 9.3 and Version 9.2 of that software.**


## **INTRODUCTION**

The macros in this paper are suitable for management reporting in business, government, or other organizational settings, not statistical or heavy-duty analytical graphs. Although the SG (Statistical Graphics) procedures were originally developed for statistical graphics, they can be used for management reporting. By graphs for management reporting I mean graphs or web graphs used to answer common questions. E.g., how do measurements for different entities compare? This question is commonly answered visually with time plots bar charts or pie charts.

All of the macros provided here are in pursuit of visual communication effectiveness.

See Reference 1 for some non-macro graphic solutions that use ODS GRAPHICS and SG procedures. Two of those solutions are made easier and less error vulnerable by use of macros provided below.

## PIE CHART WITH ORDERING AND ALL RELEVANT NUMERIC INFORMATION



There is NOTHING ELSE you could need visually or numerically to have complete understanding of the information at this level, and ordering makes it easy to quickly assess the relative size of the shares of the whole.

Within the ODS GRAPHICS environment of Base SAS, there is no SG procedure to create a pie chart. Instead, a pie chart requires use of the cumbersome, and otherwise unnecessary, GTL (Graph Template Language). Below is the code for invocation of a macro to simplify the task of creating the pie chart above:

```
options sasautos=("D:\MySASmacros" sasautos);
/* First look in D:\MySASmacros for any macro to be invoked.
   If Not Found, then look in the default sasautos,
   which is the macro library shipped by SAS Institute,
   and maybe additional local SAS macro library(ies) at your site,
   which has (have) been linked to your SAS software installer.
   You can reverse the order of the search list above.
   You can include additional custom macro libraries in the search list above.
   If the same macro name is used in multiple libraries,
   the first one found will be used. */
```

```
options mprint;
```

```
%BesslerBestPieChart27Jun2012
(Data=sashelp.shoes
,SliceLabelVar=Product
,SliceMeasureVar=Sales
,SliceMeasureFormat=dollar10.
,Order=Descending
,ChartTitle=GTL Pie Chart Created Using BesslerBestPieChart27Jun2012 Macro
,ChartFileName=PieChartCreatedUsingBesslerBestPieChart27Jun2012Macro
,ChartFolderName=D:\@WIIISU Jun2012\Results
,ChartHeight=300px
,ChartWidth=800px);
```

Here is the macro that was stored in D:\MySASmacros with filename BesslerBestPieChart27Jun2012.sas .

```
%macro BesslerBestPieChart27Jun2012
(Data=
,SliceLabelVar=
,SliceMeasureVar=
,SliceMeasureFormat=
,DecimalPositionsForPercents=1 /* can be 0 or any integer */
,Order= /* valid values are descending or ascending */
,ChartTitle=
```

```

,ChartFileName=
,ChartFolderName=
,ChartHeight=
,ChartWidth=
);

* All parameters above, except Order are mandatory *;
* If Order is not specified, SAS PROC SORT default is ascending *;

proc summary data=&Data nway;
class &SliceLabelVar;
var &SliceMeasureVar;
output out=ToPrep sum=TotalByClass;
run;

proc sql noprint;
select sum(TotalByClass) into :GrandTotal from ToPrep;
quit;

data ToChart;
length SliceNameWithPercentAndValue $ 256; /* over-sized, but that is harmless */
set ToPrep;
SliceNameWithPercentAndValue =
  trim(left(
    put(((TotalByClass / &GrandTotal) * 100),z4.&DecimalPositionsForPercents)
  )) ||
  '% - ' ||
  trim(left(&SliceLabelVar)) ||
  ' - ' ||
  trim(left(put(TotalByClass,&SliceMeasureFormat)));
run;

proc sort data=ToChart;
by &Order TotalByClass;
run;

proc template;
define statgraph BesslerBestPieChart27Jun2012;
  begingraph;
    entrytitle "&ChartTitle";
    layout region;
      piechart category=SliceNameWithPercentAndValue
               response=TotalByClass /
               datalabelcontent=(category)
               datalabellocation=callout
               otherslice=FALSE;
    endlayout;
  endgraph;
end;
run;

ods listing gpath="&ChartFolderName";

ods graphics on / reset=all
                 border=on
                 height=&ChartHeight
                 width=&ChartWidth
                 imagename="&ChartFileName";


proc sgrender data=ToChart template=BesslerBestPieChart27Jun2012;
run;

%mend BesslerBestPieChart27Jun2012;

```

## BAR CHART OF SUMS WITH PERCENT SHARES: ALTERNATIVE FOR WHEN PIE CHART IS INFEASIBLE OR UNACCEPTABLE

I have long been an advocate for horizontal bar charts rather than vertical bar charts. Vertical bar charts work well only when the bar labels are short. Tilted, or worse, vertical labels for vertical bars are somewhere between inelegant and outright anti-communicative. With V9.2, the length on labels was extended to 256, which is always adequate, and, for horizontal bar charts, always useful for longer labels that are, in fact, often needed. 256 would be impractical (no space left for the bar—unless you expand the image width, which IS possible), but it's a welcome, friendly limit.



Just as the pie chart labels included the values, not just slice names and percent shares, the bar chart labels created in this macro COULD include all three information items. I leave that macro enhancement possibility to the reader. See the code for the last macro in this paper for another way to compose multi-function bar chart labels.

The main advantage of SGPLOT bar charts is that you can get the values at the bar ends with no extraordinary effort. A PROC GCHART bar chart by default places those values in a stack or table at the right margin of the graph area. To get them at the bar ends requires use of the ANNOTATE facility, which is more work—best avoided if possible.

Below is the code for invocation of a macro to simplify the task of creating the bar chart above:

```
options sasautos=("D:\MySASmacros" sasautos);
/* First look in D:\MySASmacros for any macro to be invoked.
   If Not Found, then look in the default sasautos,
   which is the macro library shipped by SAS Institute,
   and maybe additional local SAS macro library(ies) at your site,
   which has (have) been linked to your SAS software installer.
   You can reverse the order of the search list above.
   You can include additional custom macro libraries in the search list above.
   If the same macro name is used in multiple libraries,
   the first one found will be used. */
options mprint;
%HBARsummaryChartWithPctShares
(Data=sashelp.shoes
,BarLabelVar=Product
,BarMeasureVar=Sales
,FontSize=16pt
,BarWidth=0.5
,Order=Descending
,ChartTitle=Horizontal Bar Chart of Ranked Totals and Percent Shares Using
HBARsummaryChartWithPctShares Macro
,ChartFileName=BarChartCreatedUsingHBARsummaryChartWithPctSharesMacro
,ChartFolderName=D:\@WILSU Jun2012\Results
,ChartHeight=303px /* NOTE: At 300px, every other bar label is omitted by SAS. */
,ChartWidth=800px); /* See discussion of FitPolicy=THIN in Reference 1. */
```

Here is the macro that was stored in D:\MySASmacros with filename HBARsummaryChartWithPctShares.sas .

```
%macro HBARsummaryChartWithPctShares
(Data=
,BarLabelVar=
,BarMeasureVar=
,BarMeasureFormat=
,FontSize=
,BarWidth=
,DecimalPositionsForPercents=1 /* can be 0 or any integer */
,Order= /* valid values are descending or ascending */
,ChartTitle=
,ChartFileName=
,ChartFolderName=
,ChartHeight=
,ChartWidth=
);

%if %upcase(&Order) EQ DESCENDING
%then %let Order = respdesc;
%else %let Order = respasc ;

proc summary data=&Data nway;
class &BarLabelVar;
var &BarMeasureVar;
output out=ToPrep sum=TotalByClass;
run;

proc sql noprint;
select sum(TotalByClass) into :GrandTotal from ToPrep;
quit;

data ToChart;
length BarNameWithPercent $ 256; /* over-sized, but that is harmless */
set ToPrep;
BarNameWithPercent =
    trim(left(&BarLabelVar)) ||
    '-' ||
    trim(left(put(((TotalByClass / &GrandTotal) * 100),z4.1))) ||
    '%';
run;

proc template;
define style styles.MinimalWithNoFrame; /* remove a useless box around the bars */
    parent=styles.Minimal;
    class graphwalls / frameborder=off;
end; run;

ods listing gpath="&ChartFolderName" style=styles.MinimalWithNoFrame;

ods graphics on / reset=all
                    border=on
                    height=&ChartHeight
                    width=&ChartWidth
                    imagename="&ChartFileName";

title height=&FontSize "&ChartTitle";

proc sgplot data=ToChart;
hbar BarNameWithPercent / response=TotalByClass categoryorder=&Order
                        datalabel datalabelattrs=(size=&FontSize)
                        barwidth=&BarWidth nooutline;
```

```
yaxis display=(nolabel noline noticks)
      valueattrs=(size=&FontSize);
xaxis display=none;
run;

%mend HBARsummaryChartWithPctShares;
```

## **SOLUTIONS FOR A FINITE WORK DAY IN AN ERA OF INFORMATION OVERLOAD: HOW THEY WHAT IS IMPORTANT . . . WITH THE SUBSETTED AND RANKED HORIZONTAL BAR CHART**

For over a quarter century I have advocated and exploited use of the subsetted and ranked horizontal bar chart. I always remember the wisdom of Jim White, an expert on printing, who said, "Let part stand for the whole." I actually heard him several years after I became hooked on the idea of trying to deliver only the most important. The most important can usually fit on one sheet of paper, and frequently, if not almost always, on one web page without having to scroll. I'm also reminded of the wisdom of Kenneth J. Wesley, my staff who once counseled me, when I was working on a report for executive management, that "If it doesn't fit on one page, they won't read it."

With the availability of long bar chart labels (up to 256 characters) in a recent version of SAS/GRAPH, the horizontal bar chart increased its communication power. There is no longer a need for ugly abbreviations or for appending information to bar ends with the ANNOTATE facility.

Many years ago I developed a macro that allowed the user to point to a data set and create a bar chart subsetted in any one of three ways: (1) Top N (where N was any integer); (2) all values above a cut-off; or (3) enough of the top values to account for the Top PP Percent of the total measure of interest. Whereas I have long maintained that what will fit on a page, say, anywhere from the Top 10 to the Top 40 or 50, will usually account for 80 to maybe 99% of the total measure of interest, a reliable approach to take when subsetting is to stop reporting as soon as the chart bars account as much of the total as you care about. YOU pick the percent target with Option 3 above.

However, during my latest consulting project, I decided I wanted my client's reporting system users to be able to optionally look at the WHOLE list.

The macros that I am sharing here do NOT support that four-option capability. What was delivered was actually a macro that not only created four versions of the ranking report, but also interlinked the four web graphs with hyperlinks—very cool and maximally convenient.

Here I provide a limited function macro, and only a SAS/GRAPH version, that successfully sizes the height of the image file for ANY number of the Top N bars, including ALL of the bars. It is packaged with some dummy title lines. Sizing the height of the image file must take into account title lines. Applying the macro to your needs entails using the title lines however appropriate, possibly adding or deleting title lines, and adjusting other macro parameters based on your use of titles.

Before showing results created by the macros, as well as the macros and their invocation code, let me show you a more ambitious implementation of this concept. Four ways of subsetting the ranked information are interlinked. Also, subtitles in each chart provide: (1) the count of bars presented, the subtotal of the measure of interest reported by them, and what percent of the grand total that is; and (2) the total number of bars available and their grand total of the measure of interest. The hyperlinks (which vary from case to case to omit a link to the current chart) and the numeric subtitles are dynamically generated by the macro code for this application.


This production application was built with SAS/GRAPH, but could be converted to ODS GRAPHICS and the SGPLOT procedure if an algorithm can be found to get the bar labels to reliably appear without automatic thinning of the axis values by SGPLOT. See Reference 1 for some examples of how to prevent the automatic thinning phenomenon in other situations.

# DEMO OF A PRODUCTION APPLICATION USING SUBSETTED RANKING HBAR CHARTS

All Processes Are Ranked and Listed (requires scrolling to see all of the list)

Note title lines 2 and 4


Note the hyperlinks in blue available to view other subsets



Top 50 Ranked Processes

Note title lines 2 and 4

Note the hyperlinks in blue available to view other subsets






Enough Ranked Processes To Account for Top 90% of All Read And/Or Write MB

Note title lines 2 and 4


Note the hyperlinks in blue available to view other subsets



All Ranked Processes with At Least 10,000,000 R/W Bytes

Note title lines 2 and 4


Note the hyperlinks in blue available to view other subsets




# OUTPUT EXAMPLES AND CODE FOR A SUBSETTED RANKING HBAR CHART MACRO

Adaptable Horizontal Bar GCHART Web-Deployed


Top 10 Chart




Top 25 Chart



## Top 50 Chart




## Top 100 Chart requires scrolling Top of the web page




All Of The Observations (requires scrolling)

Top of the web page

**Comment:** I have not yet found a way to eliminate the extra white space when the bar count exceeds 50. The extra space increases with bar count. Let me know if you are interested in a macro update, if this is remediated.



Bottom of the web page



Here is the macro invocation code used to create the five cases of the SAS/GRAPH PROC GCHART Adaptable Subsetted and Ranked Horizontal Bar Chart:

```
options mprint;

libname DataLib "D:\@WIILSU Jun2012\Data";

%AdaptableHorizontalBarGCHART(
Data=DataLib.PopulationByCountryPerWFB2012
,BarCountMax=10
,RptPath=D:\@WIILSU Jun2012\Results
,TitlePrefix=GCHART HBAR Chart of Ranked Population and Percent of World Total For
,TitleSuffix=Countries
,BarLabelVar=Country
,BarValueVar=Population
,BarValueFormat=comma13.);

%AdaptableHorizontalBarGCHART(
Data=DataLib.PopulationByCountryPerWFB2012
,BarCountMax=25
,RptPath=D:\@WIILSU Jun2012\Results
,TitlePrefix=GCHART HBAR Chart of Ranked Population and Percent of World Total For
,TitleSuffix=Countries
,BarLabelVar=Country
,BarValueVar=Population
,BarValueFormat=comma13.);

%AdaptableHorizontalBarGCHART(
Data=DataLib.PopulationByCountryPerWFB2012
,BarCountMax=50
,RptPath=D:\@WIILSU Jun2012\Results
,TitlePrefix=GCHART HBAR Chart of Ranked Population and Percent of World Total For
,TitleSuffix=Countries
,BarLabelVar=Country
,BarValueVar=Population
,BarValueFormat=comma13.);

%AdaptableHorizontalBarGCHART(
Data=DataLib.PopulationByCountryPerWFB2012
,BarCountMax=100
,RptPath=D:\@WIILSU Jun2012\Results
,TitlePrefix=GCHART HBAR Chart of Ranked Population and Percent of World Total For
,TitleSuffix=Countries
,BarLabelVar=Country
,BarValueVar=Population
,BarValueFormat=comma13.);

%AdaptableHorizontalBarGCHART(
Data=DataLib.PopulationByCountryPerWFB2012
,BarCountMax=All
,RptPath=D:\@WIILSU Jun2012\Results
,TitlePrefix=GCHART HBAR Chart of Ranked Population and Percent of World Total For
,TitleSuffix=Countries
,BarLabelVar=Country
,BarValueVar=Population
,BarValueFormat=comma13.);
```

This macro creates SAS/GRAPH PROC GCHART Adaptable Subsetted and Ranked Horizontal Bar Charts:

```
%macro AdaptableHorizontalBarGCHART(
Data=
,RptPath=
,TitlePrefix=
,TitleSuffix=
,BarCountMax=
,BarLabelVar=
,BarValueVar=
,BarValueFormat=
,BarColor=CXCCCCC /* light grey */
,TitleFont='Albany AMT/Bold'
,TitleHeight=16 PT
,TextFont='Albany AMT/Bold'
,TextHeight=12 PT
,X_pixels=1600
,DefaultYpixels=850
,BarCountForDefaultYpixels=50
,DefaultShareOfSpaceForBars=0.81 /* Manually adjust this based on:
                                     DefaultYpixels,
                                     BarCountForDefaultYpixels,
                                     number of TITLE lines,
                                     choice of title & text fonts,
                                     and title & text heights */
);

%if %upcase(&BarCountMax) NE ALL
%then %do;

PROC SORT DATA=&data;
BY DESCENDING &BarValueVar;
RUN;

OPTIONS OBS=&BarCountMax;

%end;

DATA ToProcess;
SET &data;
RUN;

PROC SQL NOPRINT;
SELECT COUNT(&BarLabelVar) , SUM(&BarValueVar)
      INTO :BarCount , :BarValueTotal
      FROM ToProcess;
QUIT;

OPTIONS OBS=MAX;

DATA ToPrep(KEEP=BarLabel &BarValueVar);
LENGTH BarLabel $ 256 BarLabelSuffix $ 7;
RETAIN MaxLabelLength 0;
SET ToProcess END=LastOne;
BarLabelSuffix = TRIM(LEFT(PUT(((&BarValueVar /
      &BarValueTotal) * 100),6.3))) || '%';
IF LENGTH(BarLabelSuffix) EQ 6
THEN BarLabelSuffix = '0' || BarLabelSuffix;
BarLabel =
      TRIM(LEFT(PUT(_N_,z3.))) || ' : ' ||
      TRIM(LEFT(&BarLabelVar)) || ' : ' ||
      TRIM(LEFT(PUT(&BarValueVar,&BarValueFormat)))
      || ' : ' || BarLabelSuffix;
```

```

MaxLabelLength =
    MAX(MaxLabelLength,LENGTH(BarLabel));
OUTPUT;
IF LastOne;
CALL SYMPUT('LabelLength',MaxLabelLength);
RUN;

DATA ToChart;
LENGTH BarLabel $ &LabelLength;
SET ToPrep;
RUN;

%let BarCount = %trim(%left(&BarCount));

%let Y_pixels = &DefaultYpixels;

/*

%if %eval(&BarCount NE
    &BarCountForDefaultYpixels)
%then

    NOTE: The DO group below is really necessary only if
        BarCount is not the BarCountForDefaultYpixels.
        No harm is done by running the code, however.

*/

%do;

    %put BarCount = &BarCount;

    %let DefaultYpixelsForDefaultBarCount =
    %sysfunc(CEIL( %sysevalf( &DefaultYpixels
        * &DefaultShareOfSpaceForBars ) ));

    %put DefaultYpixelsForDefaultBarCount = &DefaultYpixelsForDefaultBarCount;

    %let ReserveYpixelsForTtlsFtnsEtc =
    %eval( &DefaultYpixels
        - &DefaultYpixelsForDefaultBarCount );

    %put ReserveYpixelsForTtlsFtnsEtc = &ReserveYpixelsForTtlsFtnsEtc;

    %let YpixelsPerBar =
    %sysevalf( &DefaultYpixelsForDefaultBarCount
        / &BarCountForDefaultYpixels );

    %put YpixelsPerBar = &YpixelsPerBar;

    %let YpixelsForAllBars =
    %sysfunc(CEIL( %sysevalf(&BarCount
        * &YpixelsPerBar) ));

    %put YpixelsForAllBars = &YpixelsForAllBars;

    %let Y_pixels =
    %sysevalf(&YpixelsForAllBars
        + &ReserveYpixelsForTtlsFtnsEtc );

    %put Y_pixels = &Y_pixels;

%end;

```

```

%let Title = &TitlePrefix;

%if %upcase(&BarCountMax) EQ ALL
%then %let Title = &Title.%str( All )&BarCount;
%else %let Title = &Title.%str( Top )&BarCount;

%let Title = &Title &TitleSuffix;

%let FileName = %sysfunc(COMPRESS(&Title,' '));

GOPTIONS RESET=ALL;

GOPTIONS XPIXELS=&X_pixels YPIXELS=&Y_pixels;
GOPTIONS FTEXT=&TextFont HTEXT=&TextHeight;

* GOPTIONS BORDER; /* Uncomment this to see border of image file in web browser.
                    This is usually undesirable. */

PROC CATALOG CAT=work.gseg KILL;
RUN; QUIT;

/* The PROC step above permits reuse of a graph file name in reruns during the same
   SAS or Enterprise Guide session. However, the first time that the PROC step runs,
   it causes an ERROR message.
   For a better solution, ask for my macro Clear_GSEG_If_Present */

OPTIONS NOCENTER; FOOTNOTE;
ODS NORESULTS;
ODS LISTING CLOSE;
ODS HTML PATH="&RptPath" (URL=NONE)
  STYLE=MINIMAL GTITLE GFOOTNOTE
  BODY="&FileName..html" (TITLE="&Title");

TITLE1 HEIGHT=0.25PCT ' ' JUSTIFY=CENTER FONT=&TitleFont HEIGHT=&TitleHeight
  "&Title";
TITLE2 JUSTIFY=CENTER FONT=&TitleFont HEIGHT=&TitleHeight
  "Title Line 2";
TITLE3 JUSTIFY=CENTER FONT=&TitleFont HEIGHT=&TitleHeight
  "Title Line 3";
TITLE4 JUSTIFY=CENTER FONT=&TitleFont HEIGHT=&TitleHeight
  "Title Line 4";
TITLE5 JUSTIFY=CENTER FONT=&TitleFont HEIGHT=&TitleHeight
  "Title Line 5";

PROC GCHART DATA=ToChart;
PATTERN1 COLOR=&BarColor VALUE=SOLID;
AXIS1 LABEL=NONE MAJOR=NONE MINOR=NONE
  STYLE=0 VALUE=(JUSTIFY=RIGHT);
AXIS2 LABEL=NONE MAJOR=NONE MINOR=NONE
  STYLE=0 VALUE=NONE;
HBAR BarLabel / SUMVAR=&BarValueVar
  LEVELS=ALL DESCENDING NOSTATS NOFRAME
  MAXIS=AXIS1 RAXIS=AXIS2
  SPACE=0 COUTLINE=BLACK
  NAME="gchartHBAR&BarCount"
  DESCRIPTION=' ';
RUN; QUIT;

ODS HTML CLOSE; ODS LISTING;

%mend AdaptableHorizontalBarGCHART;

```



## CONCLUSION

ODS GRAPHICS and the SG procedures are still young technology, and users will need to wait for future upgrades before all of their obvious needs are addressed. Two of these macros can simplify the task of overcoming some current limitations of the new technology.

The SAS/GRAPH macro to create subsetting ranked horizontal bar charts implements a powerful and useful, though simple, concept and graphic method. As shown in the demo of a production application, it can be extended with even more function.

## REFERENCES

1. Bessler, LeRoy. "How To Create Communication-Effective Graphs in SAS V9.3 Without SAS/GRAPH", *Proceedings of the June 2012 Wisconsin Illinois SAS Users Conference*. Milwaukee, WI: Software User Services, 2012.  
Proceedings & Tools CD available from [Le\\_Roy\\_Bessler@wi.rr.com](mailto:Le_Roy_Bessler@wi.rr.com)

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

LeRoy Bessler PhD  
Bessler Consulting and Research  
Converting Complexity to Clarity™  
Strong Smart Systems™  
262-242-1099  
[Le\\_Roy\\_Bessler@wi.rr.com](mailto:Le_Roy_Bessler@wi.rr.com)

SAS, SAS/GRAPH, and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.  
Other brand and product names are trademarks of their respective companies.

Converting Complexity to Clarity and Strong Smart Systems are trademarks of LeRoy Bessler PhD.